

Listing Mystère

par André Chéramy

Un Mag de juillet-août se doit d'offrir quelques listings à ses lecteurs. N'ayant rien reçu de ce genre, j'ai d'abord envisagé de fouiller dans mes vieilleries. Puis je me suis souvenu d'une petite rubrique, parue il y a quelques temps dans le Ceo-Mag et dédiée à une astuce proposée par Simon Guyart. J'ai toujours eu envie d'exploiter cette astuce, mais reporté la chose faute de temps. Ce que je vous propose aujourd'hui n'est pas grandiose, mais fournit un exemple simple de ce principe, en espérant que cela vous donnera des idées de programmes 'pour de bon'. En fait je vous propose trois listings qui tournent autour de la même idée. Bien que ces exemples soient écrits en Basic, j'ai dû y mettre des WAIT, c'est vous dire la rapidité de l'astuce de Simon. Je n'ose même pas imaginer ce qu'on pourrait faire avec ça en langage machine ou en Forth. A vous de jouer et de découvrir de quoi il s'agit !

Listing n°1 : 'PETIT'

```

100 ACCENT OFF:CLS:PRINTCHR$(17)
110 CALL#F766:'Effacer CAPS
120 MOVE#B500,#B508,#B750:'Eff j
130 'Ecran avec tous les j
140 FORI=#BBA8TO#BFDFSTEP#F0
150 FORJ=ITOI+#27STEP6
160 POKEJ+#2A,#6A
170 NEXTJ
180 NEXTI
190 W=5:'Delai affichage
200 'Rotation
210 FORI=1TO15:'15 Rotations
220 MOVE#B6E0,#B6E8,#B750:'\
230 WAIT W
240 MOVE#B7E0,#B7E8,#B750:'|
250 WAIT W
260 MOVE#B578,#B580,#B750:'/'
270 WAIT W
280 MOVE#B568,#B570,#B750:'-
290 WAIT W
300 NEXTI
310 CLS:CALL#F8D0:'Régénère j
320 ACCENT SET:PRINTCHR$(17)
330 END

```

Listing n°2 : 'FLASHES'

```

100 ACCENT OFF:CLS:PRINTCHR$(17)
200 CALL#F766:'Effacer CAPS
300 GOSUB1000:'Effacer a-i
400 GOSUB2000:'Ecran + Blocs a-i
500 X=5:W=15:GOSUB4000:'Flashes
700 CLS:CALL#F8D0:'Régénère
800 PRINTCHR$(17):ACCENT SET
900 END
1000 'Effacer a-i
1010 FORZ=#B708TO#B74F
1020 POKEZ,0
1030 NEXTZ
1040 RETURN
2000 'Ecran avec Blocs a-i
2010 FORL=#BBA8TO#BFDFSTEP#F0
2020 FORC=1TO#27STEP6
2030 GOSUB3000:'Place 1 bloc a-i
2040 NEXTC
2050 NEXTL
2060 RETURN
3000 'Place un bloc a-i
3010 K=#61
3020 FORI=LTOL+#50STEP#28
3030 FORJ=CTOC+2
3040 POKEI+J,K:K=K+1
3050 NEXTJ
3060 NEXTI
3070 RETURN
4000 'Flashes
4010 FORI=1TOX
4020 GOSUB5000:'Copie a-i
4030 WAIT W
4040 GOSUB1000:'Effacer a-i
4050 NEXTI
4060 RETURN
5000 'Copie les signes sur a-i
5002 MOVE#B500,#B508,#B728:'e
5004 POKE#B72B,#08:'Point centr
5010 MOVE#B6E0,#B6E8,#B708:'a
5020 MOVE#B7E0,#B7E8,#B710:'b
5030 MOVE#B578,#B580,#B718:'c
5040 MOVE#B568,#B570,#B730:'f
5050 MOVE#B6E0,#B6E8,#B748:'i

```

```

5060 MOVE#B7E0,#B7E8,#B740:'h
5070 MOVE#B578,#B580,#B738:'g
5080 MOVE#B568,#B570,#B720:'d
5090 RETURN

```

Listing n°3 : 'RONDES'

```

100 ACCENT OFF:CLS:PRINTCHR$(17)
200 CALL#F766:'Effacer CAPS
300 GOSUB1000:'Effacer a-i
400 GOSUB2000:'Ecran + Blocs a-i
600 X=5:W=15:GOSUB6000:'Rondes
700 CLS:CALL#F8D0:'Régénère
800 PRINTCHR$(17):ACCENT SET
900 END
1000 'Effacer a-i
1010 FORZ=#B708TO#B74F
1020 POKEZ,0
1030 NEXTZ
1040 RETURN
2000 'Ecran avec Blocs a-i
2010 FORL=#BBA8TO#BFDFSTEP#F0
2020 FORC=1TO#27STEP6
2030 GOSUB3000:'Place 1 bloc a-i
2040 NEXTC
2050 NEXTL
2060 RETURN
3000 'Place un bloc a-i
3010 K=#61
3020 FORI=LTOL+#50STEP#28
3030 FORJ=CTOC+2
3040 POKEI+J,K:K=K+1
3050 NEXTJ
3060 NEXTI
3070 RETURN
6000 'Rondes
6002 FORI=1TOX
6004 POKE#B72B,#08:'Point centr
6010 MOVE#B6E0,#B6E8,#B708:'a

```

```

6012 WAIT W
6014 MOVE#B500,#B508,#B708
6016 WAIT W
6020 MOVE#B7E0,#B7E8,#B710:'b
6022 WAIT W
6024 MOVE#B500,#B508,#B710
6026 WAIT W
6030 MOVE#B578,#B580,#B718:'c
6032 WAIT W
6034 MOVE#B500,#B508,#B718
6036 WAIT W
6040 MOVE#B568,#B570,#B730:'f
6042 WAIT W
6044 MOVE#B500,#B508,#B730
6046 WAIT W
6050 MOVE#B6E0,#B6E8,#B748:'i
6052 WAIT W
6054 MOVE#B500,#B508,#B748
6056 WAIT W
6060 MOVE#B7E0,#B7E8,#B740:'h
6062 WAIT W
6064 MOVE#B500,#B508,#B740
6066 WAIT W
6070 MOVE#B578,#B580,#B738:'g
6072 WAIT W
6074 MOVE#B500,#B508,#B738
6076 WAIT W
6080 MOVE#B568,#B570,#B720:'d
6082 WAIT W
6084 MOVE#B500,#B508,#B720
6086 WAIT W
6090 NEXT
6092 RETURN

```

Vous pouvez constater la parenté entre ces trois listing, notamment entre les deux derniers, pour lesquels, vous n'aurez pas à retaper une grosse partie des lignes...

